



Programiranje kroz aplikacije

Rad sa folderima i fajlovima
Word i VBA

FileDialog objekt

- **FileDialog** objekat služi za selektovanje fajlova i foldera.
- Ovaj objekat se dobija pomoću **FileDialog** osobine objekta **Application**. Ova osobina ima jedan argument, **fileDialogType**, koji definiše tip dijaloga i ima četiri moguće vrednosti:
 - **msoFileDialogFilePicker** – dijalog za odabir fajla;
 - **msoFileDialogFolderPicker** – dijalog za odabir foldera;
 - **msoFileDialogOpen** – dijalog za otvaranje fajla;
 - **msoFileDialogSaveAs** – dijalog za snimanje fajla.
- Dakle, kreiranje **FileDialog** objekta za odabir foldera bi bilo:
`Dim FD As FileDialog`
`Set FD = Application.FileDialog(msoFileDialogFolderPicker)`


FileDialog objekt (nastavak)

■ FileDialog objekt ima:

- osobinu `InitialFileName` koja definiše početni folder,
- metodu `Show` koja prikazuje dijalog boks,
- osobinu `SelectedItems` koja daje ime selektovanog foldera.

Primer:

```
Dim FD As FileDialog
Set FD = Application.FileDialog(*folder*)
With FD
    .InitialFileName = "C:\Temp\"
    .Title = "Odaberite folder"
    .Show
End With
If FD.SelectedItems.Count = 0 Then
    MsgBox "Selekcija otkazana!"
Else
    MsgBox FD.SelectedItems(1)
End If
```



Rad sa fajlovima i folderima u VBA

Naredba	Opis
ChDir	Promena tekućeg foldera.
CurDir	Dobijanje tekućeg foldera.
ChDrive	Promena tekućeg drajva.
Dir	Dobijanje imena fajla ili foldera koje odgovara određenom obrascu, atributu fajla ili labeli drajva.
FileCopy	Kopiranje fajla.
FileDateTime	Dobijanje datuma i vremena kad je fajl poslednji put modifikovan.
FileLen	Dobijanje veličine fajla, izražene u bajtima.
GetAttr	Dobijanje atributa fajla.
Kill	Brisanje fajla.
MkDir	Kreiranje novog foldera.
Name	Promena imena fajla ili foldera.
Rmdir	Brisanje praznog foldera.
SetAttr	Promena atributa fajla.

Funkcija Dir

- Funkcija **Dir** vraća string koji predstavlja ime fajla ili foldera koje odgovara određenom obrascu. Njena sintaksa je:

ImeFajla = **Dir**(pathname, attributes)

- **pathname** – string koji određuje ime fajla (može uključiti i put),
 - **attributes** – VBA konstanta ili broj koji određuje atribut fajla. Na primer, atribut **vbDirectory** određuje pretragu foldera.
- Ako **Dir** ne pronađe traženi fajl, vraća prazan string.
 - Za specificiranje više fajlova, u funkciji **Dir** se mogu koristiti džokeri:
 - * (menja proizvoljan broj karaktera) i
 - ? (menja jedan karakter).
 - Za obilazak svih fajlova u folderu, ili fajlova određenog tipa, koristi se oblik funkcije **Dir** bez argumenata, pri čemu se pri prvom pozivu ove funkcije mora specificirati put, inače će doći do greške. Kada prođe kroz sve fajlove u folderu, **Dir** vraća prazan string. U narednom pozivu funkcije **Dir** se mora navesti put da ne bi došlo do greške. 5/41

Funkcija Dir - Primer

U aktivni radni list upisati imena svih fajlova iz određenog foldera, njihovu veličinu u bajtima, kao i datum i vreme poslednje modifikacije.

Za obilazak fajlova određenog tipa, potrebno je navesti tip fajla u prvom pozivu, npr.:
`f = Dir(Folder & "*.docx")`

```
Sub ListanjeFajlova()  
    Folder = "C:\Temp\  
    l = 1  
    With ActiveSheet  
        .Cells(l, 1) = "Ime fajla"  
        .Cells(l, 2) = "Veličina (bytes)"  
        .Cells(l, 3) = "Datum / Vreme"  
    End With  
    f = Dir(Folder)  
    Do While f <> ""  
        l = l + 1  
        With ActiveSheet  
            .Cells(l, 1) = f  
            .Cells(l, 2) = FileLen(Folder & f)  
            .Cells(l, 3) = FileDateTime(Folder & f)  
        End With  
        f = Dir  
    Loop  
End Sub
```

Funkcija Dir - Primer

```
Sub ListanjeFajlova()
```

```
Folder = "C:\Temp\"
```

```
I = 1
```

```
With ActiveSheet
```

```
.Cells(I, 1) = "Ime fajla"
```

```
.Cells(I, 2) = "Veličina (bytes)"
```

```
.Cells(I, 3) = "Datum / Vreme"
```

```
End With
```

```
f = Dir(Folder)
```

```
Do While f <> ""
```

```
I = I + 1
```

```
With ActiveSheet
```

```
.Cells(I, 1) = f
```

```
.Cells(I, 2) = FileLen(Folder & f)
```

```
.Cells(I, 3) = FileDateTime(Folder & f)
```

```
End With
```

```
f = Dir
```

```
Loop
```

```
End Sub
```

Početni folder

Zaglavlje
tabele

Za obilazak fajlova određenog tipa, potrebno je navesti tip fajla u prvom pozivu, npr.:
`f = Dir(Folder & "*.docx")`

Uzimanje prvog fajla

FileLen vraća veličinu fajla u bajtima

FileDateTime vraća datum i vreme poslednje modifikacije.

Uzimanje sledećeg fajla

Rad sa tekstualnim fajlovima

Otvaranje fajla

- Pre čitanja iz fajla, ili upisa u njega, fajl je potrebno otvoriti.
- Za otvaranje fajla se koristi VBA naredba **Open**. Mi ćemo koristiti pojednostavljeni oblik ove naredbe:

Open **pathname** **For mode** **As** **#filename**

gde je:

- **pathname** – ime fajla, uključujući put (obavezan argument);
- **mode** – način otvaranja fajla (obavezan argument) i može biti:
 - **Append** – čitanje fajla ili dopisivanje teksta na kraj fajla;
 - **Input** – čitanje fajla, bez mogućnosti upisa;
 - **Output** – čitanje i upis u fajl. Na ovaj način se kreiraju novi fajlovi, dok se brišu postojeći fajlovi sa istim imenom.
- **filename** – broj fajla (obavezan argument). U pitanju je broj između **1** i **511**; još se naziva i *handle fajla*. Radi dobijanja broja fajla koji je dostupan (trenutno nije u upotrebi), može se koristiti funkcija **FreeFile**.

Otvaranje tekst fajlova - Primeri

- Otvaranje fajla C:\Temp\Fajl.txt radi čitanja iz njega se vrši na sledeći način:

```
Open "C:\Temp\Fajl.txt" For Input As #1
```

- Ako ne znamo broj fajla koji je dostupan, možemo koristiti sledeće:

```
Dim Dostupan As Integer
```

```
Dostupan = FreeFile
```

```
Open "C:\Temp\Fajl.txt" For Input As #Dostupan
```

- Otvaranje fajla C:\Temp\Fajl.txt radi upisa u njega se vrši na sledeći način:

```
Open "C:\Temp\Fajl.txt" For Output As #1
```

Zatvaranje tekst fajlova

- Fajl koji je otvoren naredbom **Open** se zatvara naredbom **Close**, u sledećem zapisu:
Close #filenumber
- Ovom se naredbom može zatvoriti više fajlova, na sledeći način:
Close #filenumber1, #filenumber2, ..., #filenumberN
- Ukoliko se navede samo naredba **Close**, bez navođenja broja fajlova, zatvaraju se svi trenutno otvoreni fajlovi.
- Svi fajlovi otvoreni naredbom **Open** se alternativno mogu zatvoriti naredbom **Reset**.

Čitanje iz tekst fajlova

- Kod VBA postoje tri naredbe za čitanje podataka iz tekst fajlova:
 - **Input** – funkcija za čitanje određenog broja karaktera iz fajla;
 - **Input #** – čitanje podataka iz fajla i dodeljivanje tih podataka promenljivim koje su odvojene zarezima;
 - **Line Input #** – čitanje čitave linije podataka.

- Funkcija **Input** čita određeni broj karaktera iz fajla i vraća string pročitanih karaktera. Poziva se na sledeći način:

```
Procitano = Input(number, #filename)
```

gde je **number** broj karaktera koji se čita, a **filename** je handle za dati fajl. Na primer, sekvenca naredbi

```
Open "C:\temp\Fajl.txt" For Input As #1  
Procitano = Input(12, #1)
```

će pročitati prvih 12 karaktera iz fajla Fajl.txt i upisati ih u promenljivu **Procitano**.

Funkcija Input - Primer

- Napisati proceduru koja broji koliko ima cifara u fajlu Fajl.txt. Dobijeni broj prikazati pomoću message boksa.

```
Sub BrojCifaraUFajlu()  
  Dim BrCif As Integer, S As String  
  Open "C:\Temp\Fajl.txt" For Input As #1  
  BrCif = 0  
  Do Until EOF(1) ←  
    S = Input(1, #1)  
    If S Like "[0-9]" Then  
      BrCif = BrCif + 1  
    End If  
  Loop  
  MsgBox "Ima " & BrCif & " cifara"  
  Close #1  
End Sub
```

Funkcija **EOF** vraća True kada se dođe do kraja fajla i False u suprotnom.
Argument ove funkcije je handle na otvoreni fajl, naveden bez #.
Ako ne koristimo **EOF**, dolazi do greške prilikom pokušaja čitanja nakon poslednjeg karaktera.

Naredba Input

- Naredba `Input #` je pogodna ukoliko svaki red fajla sadrži fiksni broj podataka.
- `Input #` čita podatke iz otvorenog fajla i dodeljuje ih promenljivim koje su odvojene zarezima.
- Na primer, pretpostavimo da jedan red fajla čiji je handle `#1` sadrži string (naveden po navodnicima) i ceo broj (npr. "Novembar", 8). Učitavanje ova dva podatka u promenljive `S` i `X` se vrši kao:

`Input #1, S, X`

- Ilustrujmo naredbu na sledećem primeru. Pretpostavimo da svaki red fajla `Ispit.txt` sadrži ime i prezime studenta i broj poena koje je taj student dobio na ispitu. Ime i prezime su zadati u formi jednog stringa pod znacima navoda, dok je broj poena realan broj. Student je položio ispit ukoliko je dobio više od 50 poena. Napisati proceduru koja određuje i ispisuje koliko je studenata položilo ispit.

Naredba Input # - Primer

```
Sub Polozili()  
    Dim ImePrez As String, Poeni As Single, BrPol As Integer  
    Open "C:\Temp\Ispit.txt" For Input As #1  
    BrPol = 0  
    Do Until EOF(1)  
        Input #1, ImePrez, Poeni  
        If Poeni > 50 Then BrPol = BrPol + 1  
    Loop  
    MsgBox "Položilo je " & BrPol & " studenata"  
    Close #1  
End Sub
```

Naredba Line Input

- Naredba `Line Input #` čita jednu liniju teksta (do znaka za novi red) i pročitano liniju smešta u `String` promenljivu.
- Sintaksa ove naredbe je:
`Line Input #filename, varname`
gde je `varname` ime promenljive u koju smeštamo pročitano liniju.
- Dalji rad se svodi na obradu stringova.

Upis u tekst fajl. Naredba Write

- Za upis u tekstualni fajl koristimo naredbe `Write #` i `Print #`.
- Naredba `Write #` služi za upis liste vrednosti. **Upisane vrednosti su razdvojene zarezima u fajlu, dok se stringovi nalaze pod znacima navoda.**
- Podaci upisani u fajl sa `Write #` se obično čitaju naredbom `Input #` (setite se znaka navoda pri čitanju stringova sa `Input #`).
- Sintaksa naredbe `Write #` je:
`Write #filename, outputlist`
gde je `outputlist` lista promenljivih razdvojenih zarezima koje se upisuju u fajl. Umesto promenljive se može navesti i konstantna vrednost, bilo numerička, stringovna ili logička. Ukoliko se `outputlist` izostavi, upisuje se prazna linija u fajl.
- Ukoliko želimo da nakon ispisa ostanemo u istom redu, naredbu `Write #` treba završiti tačka-zarezom.

Naredba Print

- Naredba **Print #** takođe služi za upis liste vrednosti i njena sintaksa je: **Print #filename, outputlist**
gde je **outputlist** lista vrednosti koje želimo da upišemo u fajl. **Ukoliko outputlist ima više vrednosti, te vrednosti treba razdvajati tačka-zarezom. Ako se razdvajaju zarezima, u fajlu će odgovarajuće vrednosti biti razdvojene tabovima.**
- Za razliku od **Write #**, tekst je upisan onako kako smo mi naveli, bez dodavanja znaka navoda oko stringova i zareza koji razdvajaju pojedinačne unose. Ukoliko se **outputlist** izostavi, upisuje se prazna linija u fajl.
- Podaci upisani u fajl naredbom **Print #** se obično čitaju naredbama **Input** i **Line Input #**.
- Kao kod naredbe **Write #**, ako želimo da nakon ispisa ostanemo u istom redu, naredbu **Print #** treba završiti tačka-zarezom.

Rad sa tekst fajlovima - Primeri

- Napisati proceduru koja otvara tekstualni fajl C:\Fajl.txt i sve linije fajla koje sadrže reč "VBA" prepisuje u radni list. Pored pronađene linije, u radni list upisati i redni broj linije i njenu dužinu.

```
Sub Podstring()  
    Dim I As Integer, BrLin As Integer  
    Dim Linija As String  
    Open "C:\Fajl.txt" For Input As #1  
    Cells(1, 1) = "Linija"  
    Cells(1, 2) = "Broj linije"  
    Cells(1, 3) = "Dužina linije"  
    Range("A1:C1").Font.Bold = True  
    I = 1  
    BrLin = 0
```

```
Do Until EOF(1)  
    Line Input #1, Linija  
    BrLin = BrLin + 1  
    If InStr(Linija, "VBA") <> 0 Then  
        I = I + 1  
        Cells(I, 1) = Linija  
        Cells(I, 2) = BrLin  
        Cells(I, 3) = Len(Linija)  
    End If  
Loop  
Close #1  
End Sub
```

Rad sa tekst fajlovima - Primeri

- Formirati fajl **Vrednosti.txt** koji sadrži numeričke vrednosti ćelija selektovanog opsega (svaka u zasebnom redu fajla) u formatu:

vrsta: I, kolona: J, vrednost: X

gde je X vrednost ćelije u I-toj vrsti i J-toj koloni selektovanog opsega. Ne upisivati prazne ćelije i one koje ne sadrže broj.

```
Sub Vrednosti()  
  Dim I As Integer, J As Integer, Cel As Variant  
  Open "C:\Temp\Vrednosti.txt" For Output As #1  
  For I = 1 To Selection.Rows.Count  
    For J = 1 To Selection.Columns.Count  
      Cel = Selection.Cells(I, J)  
      If Cel <> "" And IsNumeric(Cel) Then  
        Print #1, "vrsta: " & I & ", kolona: " & J & ", vrednost: " & Cel  
      End If  
    Next  
  Next  
  Close #1  
End Sub
```

Šta bi se upisalo u fajl da je umesto Print korišćeno Write?

Objekt Application

- Objektni model Word-a je hijerarhijski uređen. Na vrhu modela se nalazi objekt [Application](#), koji predstavlja samu aplikaciju, tj. Word. Objekt [Application](#) sadrži druge objekte, od kojih je najčešće korišćeni objekt [Document](#), koji predstavlja otvoreni dokument.

Osobina ili metoda	Opis
Documents	Osobina koja vraća kolekciju Documents koja sadrži sve trenutno otvorene dokumente.
ActiveDocument	Osobina koja vraća aktivni dokument.
CentimetersToPoints	Metoda koja pretvara centimetre u tačke.
MillimetersToPoints	Metoda koja pretvara milimetre u tačke.
InchesToPoints	Metoda koja pretvara inče u tačke.
Quit	Metoda koja zatvara Word.

Referenciranje dokumenta

- Standardni način referenciranja je pomoću kolekcije `Documents`, i to na jedan od sledeća dva načina:

`Application.Documents("VBA.docx")`

`Application.Documents(1)`

- Ukoliko naš kod radi samo sa Word-ovim dokumentima, može se izostaviti objekat `Application` u referenciranju.
- Druga dva načina referenciranja dokumenta su
 - `ActiveDocument` → aktivni dokument
 - `ThisDocument` → dokument u kojem se nalazi VBA kod koji se trenutno izvršava
- Objekt `ActiveDocument` se dobija preko istoimene osobina objekta `Application`.

Otvaranje novog dokumenta

- Za otvaranje dokumenata se koristi metoda `Open`. U pitanju je metoda kolekcije `Documents` koja ima 10 argumenata, a obavezan je samo prvi argument – ime dokumenta sa putem.

```
Documents.Open "C:\Temp\VBA.docx"
```

- Radi lakšeg referenciranja dokumenta, često ćemo koristiti sledeći zapis:

```
Dim Dok as Document
```

```
set Dok = Documents.Open("C:\Temp\VBA.docx")
```

Kreiranje novog dokumenta

- Za kreiranje novog dokumenta se koristi metoda `Add` kolekcije `Documents`. Sintaksa (pojednostavljena) je:

```
Documents.Add(Template, NewTemplate)
```

gde su:

- `Template` – opcioni argument koji određuje šablon (template) fajl na osnovu kojeg se kreira novi dokument. `Template` se navodi kao string koji predstavlja ime .dotm fajla, zajedno sa putem do tog fajla. Ukoliko se ovaj argument izostavi, za kreiranje se koristi fajl `Normal.dotm`.
 - `NewTemplate` – opcioni argument koji određuje da li kreiramo standardni Word dokument (.docx) ili šablon (.dotm). Vrednost `True` je za .dotm, a `False` za .docx fajlove (podrazumevano).
- Mi ćemo koristiti zapis bez argumenata:
`Dim Dok as Document`
`set Dok = Documents.Add`

Document i Documents – Osobine

Osobina	Opis
Characters	Vraća kolekciju Characters - svi karakteri datog dokumenta.
Words	Vraća kolekciju Words - sve reči datog dokumenta.
Sentences	Vraća kolekciju Sentences - sve rečenice datog dokumenta.
Paragraphs	Vraća kolekciju Paragraphs - svi pasusi datog dokumenta.
Tables	Vraća kolekciju Tables - sve tabele datog dokumenta.
Comments	Vraća kolekciju Comments - svi komentari datog dokumenta.
Name	Vraća ime dokumenta.
FullName	Vraća ime dokumenta, uključujući i put do njega.
Path	Vraća put do dokumenta. Kod novih, nesnimljenih dokumenata, Path je prazan string ("").
Saved	Da li je bilo promena u dokumentu od trenutka poslednjeg snimanja. Ukoliko nije, Saved vraća True.
Content	Vraća Range objekt koji predstavlja sadržaj celog dokumenta.
Count	Vraća broj trenutno otvorenih dokumenata (kol. Documents).

Document – Metode

Metod	Opis
Activate	Aktiviranje specificiranog dokumenta, koji postaje ActiveDocument .
Close	Zatvaranje dokumenta. Metod ima (pojednostavljenu) sintaksu Close(SaveChanges) gde SaveChanges specificira da li želimo da snimimo promene pre zatvaranja. Argument SaveChanges može biti: <ul style="list-style-type: none">➤ wdSaveChanges (promene se snimaju),➤ wdDoNotSaveChanges (promene se ne snimaju) i➤ wdPromptToSaveChanges (korisnik se pita za snimanje). Ukoliko se SaveChanges izostavi, korisnik se pita za snimanje.
Save	Snimanje dokumenta.
SaveAs	Snimanje specificiranog dokumenta pod drugim imenom. Ovaj metod ima (pojednostavljenu) sintaksu: SaveAs(FileName) gde je FileName puno ime fajla, uključujući i put do njega.

Document – Metode

Metod	Opis
Undo, Redo	Vraćanje određenog broja akcija i obrnuto. Sintaksa je <code>Undo(Times)</code> i <code>Undo(Times)</code> , gde je <code>Times</code> broj akcija.
Range	Dobijanje objekta <code>Range</code> koji predstavlja tekst dokumenta određen početnom i krajnjom pozicijom. Sintaksa je <code>Range(Start, End)</code> gde su <code>Start</code> i <code>End</code> pozicija početnog i krajnjeg karaktera. Na primer: <code>ActiveDocument.Range(0,50)</code> vraća prvih 50 karaktera aktivnog dokumenta.
PrintOut	Štampanje dokumenta. Više detalja u knjizi na str. 158 ili u Help-u Word-a.

Događaji na nivou dokumenta

- Za razliku od Excel-a, gde imamo mnoštvo događaja vezanih za radne sveske i radne listove, kod Word-a imamo mali broj događaja vezanih za dokumente. Najznačajniji događaji su:

Close

New

Open

- Akcija koja okida događaj **Close** je zatvaranje dokumenta, bilo opcijom File/Close ili pozivom metode **Close**. Procedura vezana za ovaj događaj je **Document_Close** i ona se izvršava pre nego se dokument zatvori.
- Događaj **New** se odnosi samo na šablone i akcija koja okida ovaj događaj je kreiranje novog dokumenta zasnovanog na šablonu ili kad se pozove metoda **Add** i specificira se šablon.
- Akcija koja okida događaj **Open** je otvaranje dokumenta, bilo opcijom File/Open ili pozivom metode **Open**.

Objekti koji predstavljaju tekst

- Sledeći objekti omogućuju rad sa tekstom u Word VBA:

Range

Selection

Characters

Words

Sentences

Paragraphs

- U nastavku ćemo opisati svaki od ovih objekata.

Objekt Range

- U Excel-u ne postoji objekt koji predstavlja jednu ćeliju radnog lista, već se ćelija predstavlja pomoću objekta **Range**.
- U Word-u, slično, ne postoji objekt koji predstavlja jedan karakter ili reč, već se u tu svrhu koristi objekt **Range**.
- Objekt **Range** predstavlja neprekinutu sekciju teksta u dokumentu, pa može predstavljati sve između jednog karaktera i čitavog dokumenta.
- Objekt **Range** se može dobiti pomoću **metode Range** i **osobine Range**.
- **Metoda Range** se odnosi na objekt **Document** (objašnjeno - slajd 8).
- **Osobinu Range** ima veliki broj Word-ovih objekata. Ova osobina je važna jer kod nekih objekata predstavlja jedini način rada sa tekстом. Na primer, objekt **Paragraph** nema osobinu **Font**, pa se podešavanje fonta vrši preko osobine **Range**, npr.:

```
Documents(1).Paragraphs(1).Range.Font.Bold = True
```

Objekt Range – Osobine

Osobina	Opis
Bold	Vraća True ako je opseg boldovan, False ako nijedan karakter opsega nije boldovan i wdUndefined ako je deo opsega boldovan. Može se setovati na True (sve se bolduje), False (uklanja se boldovanje) ili wdToggle (True se menja u False i obrnuto).
Italic	Isto kao za osobinu Bold .
Characters	Vraća kolekciju Characters (svi karakteri opsega).
Words	Vraća kolekciju Words (sve reči opsega).
Sentences	Vraća kolekciju Sentences (sve rečenice opsega).
Paragraphs	Vraća kolekciju Paragraphs (svi pasusi opsega).
Font	Vraća ili menja Font objekt koji određuje formatiranje karaktera opsega.
Text	Vraća ili menja tekst opsega. Na primer <code>Activedocument.Paragraphs(1).Range.Text = "Prvi pasus"</code> će zameniti prvi pasus aktivnog dokumenta datim tekstom.

Objekt Range – Metode

Metoda	Opis
Copy	Kopiranje sadržaja opsega na Clipboard.
Cut	Premeštanje sadržaja opsega na Clipboard.
Paste	Smešta sadržaj Clipboard-a na tekuću poziciju opsega. Da bismo izbegli prepisivanje preko selektovanog opsega, upotrebiti metodu Collapse pre Paste .
Delete	Brisanje dela opsega ili čitavog opsega. Ako se navede bez argumenata, briše se ceo opseg. Ako se navede sa argumentima, na sledeći način: Delete(Unit, Count) briše se Count jedinica Unit , gde Unit može biti wdCharacter (podrazumevano) ili wdWord . Ako je Count pozitivan broj, brišu se jedinice unapred, a ako je negativan brisanje je unazad.

Objekt Range – Metode (nastavak)

Metoda	Opis
<code>InsertAfter</code>	Umetanje teksta posle specificiranog opsega. Sintaksa je <code>InsertAfter(Text)</code> gde <code>Text</code> predstavlja tekst koji se umeće.
<code>InsertBefore</code>	Umetanje teksta ispred specificiranog opsega. Sintaksa je <code>InsertBefore(Text)</code> gde <code>Text</code> predstavlja tekst koji se umeće.
<code>InsertParagraphAfter</code>	Umetanje karaktera za novi pasus (Enter) posle specificiranog opsega. Nakon primene metoda, opseg se proširuje tako da uključi novi pasus.
<code>InsertParagraphBefore</code>	Umetanje karaktera za novi pasus (Enter) ispred specificiranog opsega. Nakon primene metoda, opseg se proširuje tako da uključi novi pasus.
<code>Select</code>	Selektovanje specificiranog opsega.

Objekt Selection

- Objekt **Selection** predstavlja tekuću selekciju u Word-ovom dokumentu, tj. sve što je u datom trenutku selektovano (blok teksta, tekst boks, tabela, slika, ...). Ako ništa nije selektovano, **Selection** označava tekuću tačku za umetanje.
- Objekt **Selection** vraća osobina **Selection** objekta **Application**.
- U svakom trenutku u dokumentu može postojati samo jedna selekcija, a ukoliko postoji više otvorenih dokumenata, samo jedna selekcija može biti aktivna.
- Objekti **Range** i **Selection** su dosta slični. Razlikuju se po tome da **Selection** odgovara jednoj aktivnoj selekciji, a **Range** postoji nezavisno od kursora i selekcije, uvek se sastoji od teksta i može se pristupiti bilo kojem broju **Range** objekata.
- Objekti **Range** i **Selection** dele veliki broj osobina i metoda. Na sledećem slajdu je prikazan jedan broj metoda objekta **Selection** koji nisu navedeni kod objekta **Range**.

Objekt Selection – još metoda

Metod	Opis
Type	Osobina koja vraća tip selekcije.
TypeBackspace	Metod koji briše karakter koji prethodi selekciji. Ovaj metod je funkcionalno isti kao dugme Backspace.
TypeParagraph	Metod koji umeće novi, prazni, pasus. Ovaj metod je funkcionalno isti kao dugme Enter.
TypeText	Metod za umetanje teksta u dokument. Sintaksa je: TypeText(Text) gde Text predstavlja tekst koji se umeće. Na primer, ako želimo da ispred tekuće selekcije umetnemo tekst "Dobar dan", potrebno je da uradimo sledeće: Selection.Collapse wdCollapseStart Selection.TypeText "Dobar dan"

Jedan primer

Napisati proceduru koja kreira novi Word dokument kod koga je u svakom redu upisano ime jednog .docx fajla iz odabranog foldera. Folder se odabira pomoću `FileDialog` objekta.

```
Sub ImenaFajlova()  
Dim FD As FileDialog, Dok As Document  
Dim ImeFold As String, ImeFajla As String  
Set FD = Application.FileDialog( _  
    msoFileDialogFolderPicker)  
FD.InitialFileName = "C:\Temp"  
FD.Title = "Odaberite folder"  
FD.Show
```

```
If FD.SelectedItems.Count > 0 Then  
    ImeFold = FD.SelectedItems(1)  
    Set Dok = Documents.Add  
    Dok.Activate  
    ImeFajla = Dir(ImeFold & "*.docx")  
    Do While ImeFajla <> ""  
        Selection.TypeText ImeFajla  
        Selection.TypeParagraph  
        ImeFajla = Dir  
    Loop  
    Dok.SaveAs ImeFold & "\Imena.docx"  
    Dok.Close  
End If  
End Sub
```

Kolekcija Characters

- **Characters** je kolekcija svih karaktera datog objekta, bilo da je to reč, rečenica, pasus, selekcija ili ceo dokument. Na primer, `ActiveDocument.Words(3).Characters` vraća kolekciju karaktera u objektu **Range** koji predstavlja treću reč aktivnog dokumenta.
- Osobina **Characters** vraća kolekciju **Characters**. Ovu osobinu imaju:
 - Range**
 - Selection**
 - Document**
- Pristupanje pojedinačnim karakterima se vrši navođenjem rednog broja karaktera u zagradi. Na primer, naredbom `ActiveDocument.Words(1).Characters(1).Font.Color = wdColorPink` pristupamo prvom karakteru prve reči dokumenta i menjamo mu boju u ljubičastu.

Kolekcija Characters – Primer

Napisati proceduru koja u dokumentu pronalazi sve reči duže od 10 karaktera i podvlači ih.

```
Sub DuzeOd10Podvuci()  
  Dim Rec As Range  
  For Each Rec In ThisDocument.Words  
    If Rec.Characters.Count > 10 Then  
      Rec.Font.Underline = wdUnderlineSingle  
    End If  
  Next  
End Sub
```

Osobina **Count** vraća broj karaktera u datoj kolekciji.

Kolekcija Words

- **Words** je kolekcija svih reči specificiranog objekta. Na primer:

`ActiveDocument.Words`

`ActiveDocument.Sentences(2).Words`

vraćaju sve reči aktivnog dokumenta i reči druge rečenice aktivnog dokumenta, respektivno.

- Osobina **Words** predmetnog objekta vraća kolekciju **Words**. Ovu osobinu imaju objekti:

`Range`

`Selection`

`Document`

- Word VBA ne sadrži objekt koji predstavlja reč. Umesto toga, reči se klasifikuju kao objekti tipa **Range**. Samim tim, važe pomenute osobine i metode.

Kolekcija Words – Primer

Napisati proceduru koja pomoću message boksa javlja koliko reči u aktivnom dokumentu počinje velikim slovom.

```
Sub PocinjuVelikim()  
    Dim Rec As Range, Br As Integer  
    Br = 0  
    For Each Rec In ActiveDocument.Words  
        If Mid(Rec.Text, 1, 1) Like "[A-Z]" Then  
            Br = Br + 1  
        End If  
    Next  
    If Br = 0 Then  
        MsgBox "Nema reči koje počinju velikim slovom."  
    Else  
        MsgBox "Ima " & Br & " reči koje počinju velikim slovom."  
    End If  
End Sub
```

Kolekcija Sentences

- Sentences je kolekcija svih rečenica specificiranog objekta. Na primer, `ActiveDocument.Sentences`
`ActiveDocument.Paragraphs(3).Range.Sentences`
vraćaju sve rečenice aktivnog dokumenta i sve rečenice trećeg pasusa, respektivno.
- Osobina `Sentences` predmetnog objekta vraća kolekciju `Sentences`. Ovu osobinu imaju:
 - `Range`
 - `Selection`
 - `Document`
- Kao i reči, rečenice su takođe objekti tipa `Range`.
- Kao mali primer, daćemo naredbu koja bolduje poslednju rečenicu u dokumentu.
`N = ActiveDocument.Sentences.Count`
`ActiveDocument.Sentences(N).Font.Bold = True`

Kolekcija Sentences – Primer

Napisati proceduru koja sve rečenice dokumenta sa više od 10 reči boji crvenom bojom. Na kraj dokumenta dodati rečenicu koja kaže koliko ima rečenica sa više od 10 reči.

```
Sub Preko10Reci()
```

```
Dim Recenica As Range, Br As Integer
```

```
Br = 0
```

```
For Each Recenica In ActiveDocument.Sentences
```

```
    If Recenica.Words.Count > 10 Then
```

```
        Recenica.HighlightColorIndex = wdRed
```

```
        Br = Br + 1
```

```
    End If
```

```
Next
```

```
ActiveDocument.Content.Select
```

```
Selection.Collapse wdCollapseEnd
```

```
Selection.TypeParagraph
```

```
Selection.TypeText "Ima " & Br & " rečenica sa preko 10 reči."
```

```
End Sub
```

Osobina `HighlightColorIndex` vraća ili menja boju pozadine teksta opsega.